RioTinto

KMP SO_2 EEM Program – Technical Memo D01

Atmospheric Sulphur Dioxide

Method for Estimating Dry Deposition

Prepared for:

Rio Tinto Alcan
1 Smeltersite Road, P.O. Box 1800,
Kitimat, BC, Canada V8C 2H2

Prepared by:

Trent University
ERS, 1600 West Bank Drive
Peterborough, ON, Canada K9J 7B8

Submitted September 2016

Table of Contents

List of Tables

1 Overview

The total deposition of atmospheric sulphur is an essential informative indictor under the Environmental Effects Monitoring (EEM) program, e.g., it is required for the Key Performance Indicator 'critical load exceedance risk' (see ESSA et al., 2014).

Total deposition refers to the deposition of both wet and dry atmospheric sulphur species. Under the EEM, total deposition is modelled using CALPUF (ESSA et al., 2013; 2014). However, empirical observations (using passive or active samplers) provide a means to evaluate modelled atmospheric sulphur and assess changes pre- and post-modernisation. Unlike wet deposition, dry deposition measurements of sulphur are difficult to make because of the requirements for highly sophisticated methods and instrumentation (Wesely and Hicks, 2000). In general dry deposition is modelled from air concentrations of gaseous and particulate sulphur (e.g., sulphur dioxide and particulate sulphate) multiplied by dry deposition velocities estimated using modeling techniques, i.e., 'inferential' models (Vet et al., 2015).

This technical memo briefly describes the proposed method for estimating dry deposition of gaseous and particulate atmospheric sulphur species under the EEM.

2 Dry Deposition

The dry deposition flux of a gaseous or particulate species can be calculated as a product of the species' ambient atmospheric concentration and its dry deposition velocity:

$$F = C_z \times V_d$$

where F is the deposition flux, C_z is the measured ambient air concentration at height z, and V_d is the deposition velocity, which is influenced by factors such as wind speed, height of observation, heat flux, moisture availability, vegetation, and surface roughness (Wesely and Hicks, 2000).

In the resistance model approach, the total resistance to the deposition of a given substance is used to characterize the dry deposition processes (Hicks et al., 1987; Wesely and Hicks, 2000). The deposition velocity V_d (at height z) is defined as a series of three resistances (De Vos and Zhang 2012):

$$V_d^{-1} = r_a + r_b + r_c$$

Where:

r_a is the aerodynamical resistance, describing transfer of the gaseous substance in the atmospheric surface layer,

r_b is the quasi-laminar resistance, describing transfer of the gas through the quasi-laminar layer in contact with the surface, and

r_c is the surface resistance, describing the absorption of the substance by the surface (soil or canopy). Surface resistance can further be decomposed into two parallel resistances: a stomatal resistance and a non-stomatal resistance. One of the most widely used V_d models (i.e., Zhang et al., 2001, 2003a) uses a two-big-leaf approach[1] to estimate stomatal resistance.

The V_d model developed by Zhang[2] (Zhang et al., 2001, 2003a), sometimes referred to as the CAPMoN or Environment Canada model, has been compared to American (e.g., CASTNET) and European dry deposition models (see Schwede et al., 2011, and Flechard et al., 2011), and also specifically evaluated and parameterised for sulphur dioxide (Zhang et al., 2003b). The model is widely used (e.g., see De Vos and Zhang, 2012 [Belgium], Flechard et al., 2011 [65 sites across Europe], and Yu et al., 2013 [Beijing]), with the initial model descriptions in Zhang et al. (2001) cited more than 500 times, and Zhang et al. (2003a) cited more than 250 times. The model is routinely used by Environment and Climate Change Canada to estimate dry deposition at air and precipitation stations in their ‘CAPMoN’ monitoring network (Schwede et al., 2011), also embedded into their continental-scale atmospheric deposition model (Zhang et al., 2002).

Under the EEM, it is proposed that V_d for gaseous and particulate sulphate species be calculated using the model of Zhang et al. (2001, 2003a); for a complete description of the model see Zhang et al. (2001, 2003a) and Zhang and He (2014). In addition, see Zhang et al. (2003b) for an evaluation of a non-stomatal resistance parameterization for sulphur dioxide. It is proposed that the model (available as FORTRAN code [see Appendix]) be used to calculate hourly V_d at each specific monitoring site, which can then be combined with continuous data from the monitoring stations to estimate the dry deposition of sulphur dioxide, or summarised for longer periods and combined with data from passive samplers.

3 Model Inputs

The Zhang et al. (2003) model requires two sets of input data, site specific variables and meteorological forcing variables. Three site-specific variables are required: the reference height (m), the fraction of land cover types (26 categories see Table 1) and latitude for the study site. In general, dry deposition estimates are limited to the specific monitoring location (i.e., the location where measurements are carried out), or the area represented by the measurements (i.e., a defined area surrounding the measurement location). Thirteen meteorological forcing variables on an hourly resolution for the period of interest are required. The data may be obtained from local meteorological stations or where on-site observations are unavailable; data may be obtained[3] from the Global Environmental Multiscale (GEM) meteorological model (Coté et al., 1997), i.e., Canada’s weather forecast model. Data from GEM are routinely used to estimate dry deposition at the CAPMoN stations (Schwede et al., 2011). The required meteorological forcing variables are:

Year
Month
Day

[1] Big-leaf models treat the canopy as single big leaf by mapping properties of the whole canopy onto a single leaf. Big-leaf models require parameters at the canopy level that cannot be measured directly, nor defined as the arithmetic mean of leaf-level parameters because of nonlinearity. Rather, they require some plausible assumption about the vertical profile of leaf properties.
[2] Leming Zhang, Environment and Climate Change Canada, has >100 publications, with the majority focused on dry deposition modelling and assessment. His four top cited papers focus on the V_d model, and have been cited more than 1000 times. URL: scholar.google.ca/citations?hl=en&user=qvV5wUEAAAAJ
[3] Data must be requested from Environment and Climate Change Canada.

Hour (LST)
Temperature at Z2 (K)
Surface temperature (K)
Windspeed at Z2 (m s^{-1})
RH fraction (0–1)
Solar irradiance (w m^{-2})
Precipitation rate (mm hr^{-1})
Surface pressure (mb)
Snow depth (cm)
Cloud fraction (0–1)

Table 1. Land use classes (LUC) included in the Zhang et al. (2003) deposition velocity model.

LUC	Vegetation type	LUC	Vegetation type
1	Water	14	Long grass
2	Ice	15	Crops
3	Inland lake	16	Rice
4	Evergreen needleleaf trees	17	Sugar
5	Evergreen broadleaf trees	18	Maize
6	Deciduous needleleaf trees	19	Cotton
7	Deciduous broadleaf trees	20	Irrigated crops
8	Tropical broadleaf trees	21	Urban
9	Drought deciduous trees	22	Tundra
10	Evergreen broadleaf shrub	23	Swamp
11	Deciduous shrubs	24	Desert
12	Thorn shrubs	25	Mixed wood forests
13	Short grass and forbs	26	Transitional forest

4 Literature Cited

ESSA Technologies, J. Laurence, Limnotek, Risk Sciences International, Rio Tinto Alcan, Trent University, Trinity Consultants and University of Illinois. 2013. Sulphur Dioxide Technical Assessment Report in Support of the 2013 Application to Amend the P2-00001 Multimedia Permit for the Kitimat Modernization Project. Vol.2: Final Technical Report. Prepared for RTA, Kitimat, BC. 450 pp.

ESSA Technologies, J. Laurence, Risk Sciences International, Trent University, and Trinity Consultants. 2014. Kitimat Airshed Emissions Effects Assessment. Report prepared for BC Ministry of Environment, Smithers, BC. 205 pp. + appendices.

Coté, J., J.G. Desmarais, S. Gravel, A. Methot, A. Patoine, M. Roch and A. Staniforth. 1997. The operational CMC/MRB global environmental multiscale (GEM) model, Part I, design considerations and formulation, Monthly Weather Review 126, 1373–1395.

De Vos, T. and L. Zhang. 2012. High resolution mapping of total deposition of acidifying pollutants. Atmospheric Environment 57 80–90.

Flechard, C.R., E. Nemitz, R.I. Smith, D. Fowler, A.T. Vermeulen, A. Bleeker, J.W. Erisman, D. Simpson, L. Zhang, Y.S. Tang, and M.A. Sutton, 2011. Dry deposition of reactive nitrogen to European ecosystems: a comparison of inferential models across the NitroEurope network. Atmospheric Chemistry and Physics 11, 2703–2728

Hicks, B., D. Baldocchi, T. Meyers, T. Hosker Jr. and D. Matt. 1987. A preliminary multiple resistance routine for deriving dry deposition velocities from measured quantities. Water, Air and Soil Pollution 36, 311–330.

Schwede, D., L. Zhang, R. Vet and G. Lear. 2011. An intercomparison of the deposition models used in the CASTNET and CAPMoN networks. Atmospheric Environment 45:6, 1337–1346.

Vet, R., R.S. Artz, S. Carou, M. Shaw, C.-U. Ro, W. Aas, and others. 2014. A global assessment of precipitation chemistry and deposition of sulfur, nitrogen, sea salt, base cations, organic acids, acidity and pH, and phosphorus. Atmospheric Environment 93, 3–100.

Wesely, M.L. and B.B. Hicks. 2000. A review on current status of knowledge on dry deposition. Atmospheric Environment 34, 2261–2282.

Yu L., G. Wang, R. Zhang, L. Zhang, Y. Song and B. Wu. 2013. Characterization and source apportionment of $PM_{2.5}$ in an urban environment in Beijing. Aerosol and Air Quality Research 13, 574–583.

Zhang, L.M., S.L. Gong, J. Padro and L. Barrie. 2001. A size-segregated particle dry deposition scheme for an atmospheric aerosol module, Atmospheric Environment 35, 549–560.

Zhang L., M.D. Moran, P.A. Makar, J.R. Brook and S. Gong. 2002.Modelling gaseous dry deposition in AURAMS: a unified regional air-quality modelling system. Atmospheric Environment 36:3, 537–560.

Zhang, L., J.R. Brook and R.Vet. 2003a. A revised parameterization for gaseous dry deposition in air-quality models, Atmospheric Chemistry and Physics 3, 2067–2082.

Zhang, L., J.R. Brook and R. Vet. 2003b. Evaluation of a non-stomatal resistance parameterization for SO2 dry deposition. Atmospheric Environment, 37:21, 2941–2947.

Zhang, L. and Z. He. 2014. Technical Note: An empirical algorithm estimating dry deposition velocity of fine, coarse and giant particles. Atmospheric Chemistry and Physics 14, 3729–3737.

5 Appendix

```
C    Calculate dry deposition velocities for 31 gas species
C    and 7 aerosol species
C
C----------------------------------------------------------------------

      IMPLICIT REAL(A-H,O-Z),INTEGER(I-N)
      PARAMETER (ng=31)        ! NUMBER OF GAS SPECIES DRY DEPOSITED
      PARAMETER (luc=26)       ! NUMBER OF LAND-USE CATEGORIES
      PARAMETER (NSIZE=13)     !number of size bins
      PARAMETER (np=14)        ! NUMBER OF particle species

      REAL Ra(LUC)
      REAL Z01(LUC), Z02(LUC),  ustar(LUC)
      REAL LAI(LUC,15),LAI_F(LUC)
C
C  paramaters for gaseous Vd submoudle
C
      REAL VDG(LUC,NG), ALPHA(NG),BETA(NG),RM(NG),MW(NG)
      REAL Rac1(LUC),   Rac2(LUC), RcutdO(LUC), RcutwO(LUC),
     &   RcutdS(LUC),   RgS(LUC),  RgO(LUC),   SDmax(LUC),
     &    Tmin(LUC),    Tmax(LUC),   TOPT(LUC),    BVPD(LUC),
     &    PSI1(LUC),    PSI2(LUC),  RSmin(LUC),     BRS(LUC)
      LOGICAL is_dew, is_rain
C
C  external functions
C
      external amin1, amax1

C
C   Surface Roughness Length [m].
C   Z01 and Z02 are minimum and maximum z0 for each LUC.
C
      DATA   Z01             /
     &  0.0 ,  0.01,  0.0 ,  0.9 ,  2.0 ,
     &  0.4 ,  0.4 ,  2.5 ,  0.6 ,  0.2 ,
     &  0.05,  0.2 ,  0.04,  0.02,  0.02,
     &  0.02,  0.02,  0.02,  0.02,  0.05,
     &  1.0 ,  0.03,  0.1 ,  0.04,  0.6 ,
     &  0.6    /
      DATA   Z02             /
     &  0.0 ,  0.01,  0.0 ,  0.9 ,  2.0 ,
     &  0.9 ,  1.0 ,  2.5 ,  0.6 ,  0.2 ,
     &  0.2 ,  0.2 ,  0.04,  0.1 ,  0.1 ,
     &  0.1 ,  0.1 ,  0.1 ,  0.2 ,  0.05,
     &  1.0 ,  0.03,  0.1 ,  0.04,  0.9 ,
     &  0.9    /
C
C   In-canopy aerodynamic resistance  [s/m].
C   Rac1 and Rac2 are minimum and maximum Rac0 for each LUC.
C
      DATA   Rac1            /
     &  0   ,  0   ,  0   ,  100 ,  250 ,
     &  60  ,   60 ,  300 ,  100 ,  60  ,
```

```
     &  20   ,  40   ,  20   ,  10   ,  10   ,
     &  10   ,  10   ,  10   ,  10   ,  20   ,
     &  40   ,  0    ,  20   ,  0    ,  100  ,
     &  100     /
      DATA   Rac2             /
     &  0    ,  0    ,  0    ,  100  ,  250  ,
     &  100  ,  100  ,  300  ,  100  ,  60   ,
     &  60   ,  40   ,  20   ,  40   ,  40   ,
     &  40   ,  40   ,  50   ,  40   ,  20   ,
     &  40   ,  0    ,  20   ,  0    ,  100  ,
     &  100     /
C
C    Dry cuticle resistance for SO2  [s/m].
C
      DATA   RcutdS           /
     & -999 , -999 , -999 , 2000 , 2500 ,
     & 2000 , 2500 , 2500 , 6000 , 2000 ,
     & 2000 , 2000 , 1000 , 1000 , 1500 ,
     & 1500 , 2000 , 2000 , 2000 , 2000 ,
     & 4000 , 2000 , 1500 , -999 , 2500 ,
     & 2500     /
C
c    & -999 , -999 , -999 , 1500 , 2500 ,
c    & 1000 , 2000 , 2000 , 6000 , 2000 ,
c    & 2000 , 2000 , 1000 , 1000 , 1500 ,
c    & 1000 , 2000 , 2000 , 2000 , 2000 ,
c    & 4000 , 2000 , 1000 , -999 , 2000 ,
c    & 2000     /
C
C    Ground resistance for SO2  [s/m].
C
      DATA   RgS             /
     &   20 ,   70 ,  20  ,  200 ,  100 ,
     &  200 ,  200 ,  100 ,  300 ,  200 ,
     &  200 ,  200 ,  200 ,  200 ,  200 ,
     &   50 ,  200 ,  200 ,  200 ,   50 ,
     &  300 ,  300 ,   50 ,  700 ,  200 ,
     &  200     /
C
C    Stomatal resistance related parameters.
C    In sequence: rsmin, brs, tmin, tmax, topt, bvpd, psi1, psi2
C
      DATA   rsmin            /
     & -999 , -999 , -999 ,  250 ,  150 ,
     &  250 ,  150 ,  150 ,  250 ,  150 ,
     &  150 ,  250 ,  150 ,  100 ,  120 ,
     &  120 ,  120 ,  250 ,  125 ,  150 ,
     &  200 ,  150 ,  150 , -999 ,  150 ,
     &  150    /
      DATA   brs              /
     & -999 , -999 , -999 ,   44 ,   40 ,
     &   44 ,   43 ,   40 ,   44 ,   40 ,
     &   44 ,   44 ,   50 ,   20 ,   40 ,
     &   40 ,   50 ,   65 ,   65 ,   40 ,
     &   42 ,   25 ,   40 , -999 ,   44 ,
     &   43    /
      DATA   tmin             /
```

```
     & -999 , -999 , -999 ,   -5 ,    0 ,
     &   -5 ,    0 ,    0 ,    0 ,    0 ,
     &   -5 ,    0 ,    5 ,    5 ,    5 ,
     &    5 ,    5 ,    5 ,   10 ,    5 ,
     &    0 ,   -5 ,    0 , -999 ,   -3 ,
     &    0   /
      DATA   tmax               /
     & -999 , -999 , -999 ,   40 ,   45 ,
     &   40 ,   45 ,   45 ,   45 ,   45 ,
     &   40 ,   45 ,   40 ,   45 ,   45 ,
     &   45 ,   45 ,   45 ,   45 ,   45 ,
     &   45 ,   40 ,   45 , -999 ,   42 ,
     &   45   /
      DATA   topt               /
     & -999 , -999 , -999 ,   15 ,   30 ,
     &   15 ,   27 ,   30 ,   25 ,   30 ,
     &   15 ,   25 ,   30 ,   25 ,   27 ,
     &   27 ,   25 ,   25 ,   30 ,   25 ,
     &   22 ,   20 ,   20 , -999 ,   21 ,
     &   25   /
      DATA   bvpd               /
     & -999 , -999 , -999 ,  0.31,  0.27,
     &  0.31,  0.36,  0.27,  0.31,  0.27,
     &  0.27,  0.27,  0.0 ,  0.0 ,  0.0 ,
     &  0.0 ,  0.0 ,  0.0 ,  0.0 ,  0.0 ,
     &  0.31,  0.24,  0.27, -999 ,  0.34,
     &  0.31   /
      DATA   psi1              /
     & -999 , -999 , -999 , -2.0 , -1.0 ,
     & -2.0 , -1.9 , -1.0 , -1.0 , -2.0 ,
     & -2.0 , -2.0 , -1.5 , -1.5 , -1.5 ,
     & -1.5 , -1.5 , -1.5 , -1.5 , -1.5 ,
     & -1.5 ,    0 , -1.5 , -999 , -2.0 ,
     & -2.0    /
      DATA   psi2              /
     & -999 , -999 , -999 , -2.5 , -5.0 ,
     & -2.5 , -2.5 , -5.0 , -4.0 , -4.0 ,
     & -4.0 , -3.5 , -2.5 , -2.5 , -2.5 ,
     & -2.5 , -2.5 , -2.5 , -2.5 , -2.5 ,
     & -3.0 , -1.5 , -2.5 , -999 , -2.5 ,
     & -3.0    /
C
C Leaf area index at the beginning of each month (im=1,13),
C  minimum LAI (im=14) and maximum LAI (im=15).
C Values of LAI are from GEM with modifications for urban.
C
      DATA (LAI(6,im), im = 1, 15)/
     & 0.1  , 0.1  , 0.5  , 1.0  , 2.0  ,
     & 4.0  , 5.0  , 5.0  , 4.0  , 2.0  ,
     & 1.0  , 0.1  , 0.1  , 0.1  , 5.0  /
      DATA (LAI(7,im), im = 1, 15)/
     & 0.1  , 0.1  , 0.5  , 1.0  , 2.0  ,
     & 4.0  , 5.0  , 5.0  , 4.0  , 2.0  ,
     & 1.0  , 0.1  , 0.1  , 0.1  , 5.0  /
      DATA (LAI(11,im), im = 1, 15)/
     & 0.5  , 0.5  , 1.0  , 1.0  , 1.5  ,
     & 2.0  , 3.0  , 3.0  , 2.0  , 1.5  ,
```

```
      & 1.0  , 0.5  , 0.5  , 0.5  , 3.0  /
       DATA (LAI(14,im), im = 1, 15)/
      & 0.5  , 0.5  , 0.5  , 0.5  , 0.5  ,
      & 0.5  , 1.0  , 2.0  , 2.0  , 1.5  ,
      & 1.0  , 1.0  , 0.5  , 0.5  , 2.0  /
       DATA (LAI(15,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  ,
      & 2.0  , 3.0  , 3.5  , 4.0  , 0.1  ,
      & 0.1  , 0.1  , 0.1  , 0.1  , 4.0  /
       DATA (LAI(16,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  ,
      & 2.5  , 4.0  , 5.0  , 6.0  , 0.1  ,
      & 0.1  , 0.1  , 0.1  , 0.1  , 6.0  /
       DATA (LAI(17,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  ,
      & 3.0  , 4.0  , 4.5  , 5.0  , 0.1  ,
      & 0.1  , 0.1  , 0.1  , 0.1  , 5.0  /
       DATA (LAI(18,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  ,
      & 2.0  , 3.0  , 3.5  , 4.0  , 0.1  ,
      & 0.1  , 0.1  , 0.1  , 0.1  , 4.0  /
       DATA (LAI(19,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.5  , 1.0  ,
      & 3.0  , 4.0  , 4.5  , 5.0  , 0.1  ,
      & 0.1  , 0.1  , 0.1  , 0.1  , 5.0  /
       DATA (LAI(21,im), im = 1, 15)/
      & 0.1  , 0.1  , 0.1  , 0.1  , 0.5  ,
      & 1.0  , 1.0  , 1.0  , 1.0  , 1.0  ,
      & 0.4  , 0.1  , 0.1  , 0.1  , 1.0  /
       DATA (LAI(22,im), im = 1, 15)/
      & 1.0  , 1.0  , 0.5  , 0.1  , 0.1  ,
      & 0.1  , 0.1  , 1.0  , 2.0  , 1.5  ,
      & 1.5  , 1.0  , 1.0  , 0.1  , 2.0  /
       DATA (LAI(25,im), im = 1, 15)/
      & 3.0  , 3.0  , 3.0  , 4.0  , 4.5  ,
      & 5.0  , 5.0  , 5.0  , 4.0  , 3.0  ,
      & 3.0  , 3.0  , 3.0  , 3.0  , 5.0  /
       DATA (LAI(26,im), im = 1, 15)/
      & 3.0  , 3.0  , 3.0  , 4.0  , 4.5  ,
      & 5.0  , 5.0  , 5.0  , 4.0  , 3.0  ,
      & 3.0  , 3.0  , 3.0  , 3.0  , 5.0  /
C
C  Gas Properties (Total 31 species)
C
C  Mesophyll resistance RM, scaling factors ALPHA and BETA
C
       DATA   RM             /
      & 0.   , 0.   , 0.   ,  0.  ,  0.   ,
      & 0.   , 0.   , 0.   ,  0.  ,  0.   ,
      & 0.   , 0.   , 0.   ,  0.  ,  100.,
      & 100. , 100. , 100. ,  100.,  0.   ,
      & 100. , 0.   , 0.   ,  0.  ,  0.   ,
      & 0.   , 0.   , 0.   ,  0.  ,  100.,
      & 100.     /

       DATA   ALPHA          /
      & 1.   , 1.   , 0.   ,  0.  ,  1.   ,
```

```
     & 10.  , 2.   , 5.   ,  1.  ,  0.  ,
     & 0.   , 0.   , 0.   ,  0.8 ,  0.  ,
     & 0.   , 0.   , 0.   ,  0.  ,  0.  ,
     & 0.   , 0.01 , 0.6  ,  0.6 ,  0.4 ,
     & 0.01 , 2.   , 1.5  ,  0.1 ,  0.  ,
     & 0.        /

      DATA   BETA            /
     & 0.   , 1.   , 0.8  ,  1.  ,  1.  ,
     & 10.  , 2.   , 5.   ,  0.0 ,  0.6 ,
     & 0.6  , 0.8  , 0.3  ,  0.2 ,  0.05,
     & 0.05 , 0.05 , 0.05 ,  0.05,  0.05,
     & 0.05 , 0.   , 0.1  ,  0.  ,  0.  ,
     & 0.   , 0.   , 0.   ,  0.8 ,  0.5 ,
     & 0.5       /

      DATA   MW            /
     & 64.  , 98.  , 46.  ,  48. ,  34. ,
     & 63.  , 47.  , 79.  ,  17. ,  121.,
     & 135. , 183. , 147. ,  30. ,  44. ,
     & 58.  , 72.  , 128. ,  106.,  70. ,
     & 70.  , 72.  , 32.  ,  46. ,  60. ,
     & 104. , 46.  , 60.  ,  48. ,  77. ,
     & 147.      /
C
C  Maximum snow depth over which snow fraction for leaves is 1.0
C  Snow fraction for ground is treated 2 times of that for leaves
C
      DATA SDMAX  /
     &              9999. , 1.0   , 9999. ,  200. ,  400. ,
     &               200. , 200.  ,  400. ,  200. ,   50. ,
     &                50. ,  50.  ,    5. ,   20. ,   10. ,
     &                10. ,  10.  ,  10.  ,   10. ,   10. ,
     &                50. ,   2.  ,  10.  ,    2. ,  200. ,
     &               200.        /

C --- parameters for air dynamic properties
      DATA AA1/1.257/, AA2/0.4/, AA3/1.1/
      DATA AMFP /6.53E-8/, ROAROW/1.19/,
     &    BOLTZK/1.3806044503487214E-23/
C
C --- Define the function for saturation vapor pressure (mb)
C
      ES(TEMP) = 6.108*EXP(17.27*(TEMP - 273.16)/(TEMP - 35.86))
C
C   Some constants
C
      dair=0.369*29.+6.29
      dh2o=0.369*18.+6.29

C   Initialize Leaf Area Index for LUC with constant LAI values
      DO im=1, 15
      LAI(1,im)=0.
      LAI(2,im)=0.
      LAI(3,im)=0.
      LAI(4,im)=5.
      LAI(5,im)=6.
```

```
      LAI(8,im)=6.
      LAI(9,im)=4.
      LAI(10,im)=3.
      LAI(12,im)=3.
      LAI(13,im)=1.
      LAI(20,im)=1.
      LAI(23,im)=4.
      LAI(24,im)=0.
      END DO

      VDG=0.

C
C Loop 200 for LUC
C
      DO 200 I=1,LUC

C
C interpolate LAI
C
           IM = INT(iday / 30.5 ) + 1
           iday_M =iday - INT((IM-1)*30.5+0.5)
           IF (iday_M.EQ.0) THEN
           IM=IM-1
           iday_M =iday - (IM-1)*30.5
           END IF

           LAI_F(I)  = LAI(I,IM)
     &          + iday_M / 30.5 * (LAI(I,IM+1)-LAI(I,IM))
C
C    Set minimum wind speed as 0.8 m/s
           U2 = amax1(U2,0.8)
C    Potential temperature at reference height Z2
           T2P = T2 + Z2 * 0.0098
C
C calculating friction velocity and stability related varibles
C
C for water surfaces (LUC 1, 3), z0 is a function of wind speed
C
      IF(I.EQ.1.OR.I.EQ.3)  THEN
             E = RH * ES(T2)
             Q = 0.622 * E/(PMB-E)
             T2PV = T2P * (1. + 0.61 * Q)
             E = ES(TS)
             QS = 0.622 * E / (PMB - E)
             TSV = TS * (1. + 0.61 * QS)
             DTHV = (T2PV - TSV)
             CUN=7.5E-4+6.7E-5*U2
             EL=9999.
             IF(ABS(DTHV).GT.1.0E-6)
     &       EL=T2PV*CUN**1.5*U2**2/(5.096E-3*DTHV)
             IF(EL.GT.0..AND.EL.LT.5.0) EL = 5.0
             IF(EL.GT.-5.0.AND.EL.LT.0) EL =-5.0
             ZL = Z2/EL

             IF(ZL.LT.0.0) THEN
                X=(1.0 - 15.0*ZL)**0.25
```

```
                PSIU=2.*ALOG(0.5*(1.0+X))+ALOG(0.5*(1.0+X*X))-
     &                  2.0*ATAN(X) + 0.5*3.1415926
                Y=SQRT(1.-9.*ZL)
                PSIT=2.*0.74*ALOG((1+Y)/2.)
             ELSE
                PSIU = -4.7*ZL
                PSIT = PSIU
             ENDIF

             Z0_F = 0.000002 * U2**2.5
             USTAR(I) = 0.4*U2/(ALOG(Z2/Z0_F) - PSIU)
            THSTAR = 0.4*(T2P-TS)/(0.74*ALOG(Z2/Z0_F)-PSIT)
c
C   for LUC other than water
C
      ELSE

             IF (Z02(I).GT.Z01(I)) THEN
              Z0_F=Z01(I)+(LAI_F(I)-LAI(I,14))/(LAI(I,15)-LAI(I,14))
     &                        *(Z02(I)-Z01(I))
             ELSE
              Z0_F=Z01(I)
             END IF
             RIB = 9.81*Z2*(T2P - TS)/(TS*U2**2)
             IF (SRAD.GT.0.0.AND.RIB.GT.0.0) RIB = 1.E-15
             DELTAT = T2P - TS
             IF(ABS(DELTAT).LT.1.E-10) DELTAT=SIGN(1.E-10,DELTAT)
             TBAR = 0.5*(T2P + TS)
             RATIOZ = Z2/Z0_F
             ASQ = 0.16/(ALOG(RATIOZ))**2

             IF(RIB.LE.0.)THEN
               AA = ASQ*9.4*SQRT(RATIOZ)
               CM = 7.4*AA
               CH = 5.3*AA
               FM = 1. - (9.4*RIB/(1. + CM*SQRT(ABS(RIB))))
               FH = 1. - (9.4*RIB/(1. + CH*SQRT(ABS(RIB))))
             ELSE
               FM = 1./((1. + 4.7*RIB)**2)
               FH = FM
             ENDIF

              USTARSQ = ASQ*U2**2*FM
              UTSTAR = ASQ*U2*DELTAT*FH/0.74
              USTAR(I) = SQRT(USTARSQ)
              THSTAR = UTSTAR/USTAR(I)
              EL = TBAR*USTARSQ/(0.4*9.81*THSTAR)
             IF(EL.GT.0..AND.EL.LT.5.0) EL = 5.0
             IF(EL.GT.-5.0.AND.EL.LT.0) EL =-5.0
              ZL=Z2/EL
      ENDIF
C
C     Aerodynamic resistance above canopy
C
      IF(ZL.GE.0.) THEN
          Ra(I)=(.74*ALOG(Z2/Z0_F)+4.7*ZL)/0.4/USTAR(I)
      ELSE
```

```
          Ra(I)=0.74/0.4/USTAR(I)*(ALOG(Z2/Z0_F)-
     &          2*ALOG((1+SQRT(1-9.*ZL))*0.5))
      ENDIF

      Ra(I)=amax1(Ra(I),5.0)
      if (I.EQ.1.OR.I.EQ.3) THEN
          Ra(I)=amin1(Ra(I),2000.)
      else
          Ra(I)=amin1(Ra(I),1000.)
      end if

C
C --- STOMATAL RESISTANCE FOR WATER VAPOR ONLY. STEPS FOR CALCULATING:
C
      IF ( SRAD.GE.0.1              .AND.
     &     TS.LT.(Tmax(I)+273.15)   .AND.
     &     TS.GT.(Tmin(I)+273.15)   .AND.
     &     LAI_F(I).GT.0.001            .AND.
     &     COSZEN.GT.0.001                ) THEN

C --  Calculate direct and diffuse PAR from solar radiation and solar
zenith angle

      RDU=600.*EXP(-0.185/COSZEN)*COSZEN
      RDV=0.4*(600.-RDU)*COSZEN
      WW=-ALOG(COSZEN)/2.302585
      WW=-1.195+0.4459*WW-0.0345*WW**2
      WW=1320*10**WW
      RDM=(720.*EXP(-0.06/COSZEN)-WW)*COSZEN
      RDN=0.6*(720-RDM-WW)*COSZEN
      RV=amax1(0.1,RDU+RDV)
      RN=amax1(0.01,RDM+RDN)
      RATIO=amin1(0.9,SRAD/(RV+RN))
      SV=RATIO*RV                           ! Total PAR
      FV=amin1(0.99, (0.9-RATIO)/0.7)
      FV=amax1(0.01,RDU/RV*(1.0-FV**0.6667)) ! Fraction of PAR
      PARDIR=FV*SV                          ! PAR from direct radiation
      PARDIF=SV-PARDIR                      ! PAR from diffuse radiation
c
C -- Calculate sunlit and shaded leaf area, PAR for leaves
C
      IF (LAI_F(I).GT.2.5.AND.SRAD.GT.200.) THEN
      PSHAD=PARDIF*EXP(-0.5*LAI_F(I)**0.8)
     & +0.07*PARDIR*(1.1-0.1*LAI_F(I))*EXP(-COSZEN)
      PSUN=PARDIR**0.8*.5/COSZEN+PSHAD
      ELSE
      PSHAD=PARDIF*EXP(-0.5*LAI_F(I)**0.7)
     & +0.07*PARDIR*(1.1-0.1*LAI_F(I))*EXP(-COSZEN)
      PSUN=PARDIR*.5/COSZEN+PSHAD
      END IF
      RSHAD=RSmin(I)+BRS(I)*RSMIN(I)/PSHAD
      RSUN=RSmin(I)+BRS(I)*RSMIN(I)/PSUN
      GSHAD=1./RSHAD
      GSUN=1./RSUN
      FSUN=2*COSZEN*(1.-EXP(-0.5*LAI_F(I)/COSZEN))   ! Sunlit leaf area
      FSHAD=LAI_F(I)-FSUN                   ! Shaded leaf area
```

```
C -- Stomatal conductance before including effects of temperature,
C                    vapor pressure deficit and water stress.

      GSPAR=FSUN*GSUN+FSHAD*GSHAD

C --  function for temperature effect
      T=TS-273.15
      BT=(Tmax(I)-TOPT(I))/(TOPT(I)-Tmin(I))
      GT=(Tmax(I)-T)/(TMAX(I)-TOPT(I))
      GT=GT**BT
      GT=GT*(T-Tmin(I))/(TOPT(I)-TMIN(I))
C --  function for vapor pressure deficit
      D0= ES(TS)*(1.- RH)/10.           !kPa
      GD=1.-BVPD(I)*D0
C --  function for water stress
      PSI=(-0.72-0.0013*SRAD)
c      PSI_S=(-0.395-0.043*(TS-273.15))*102.
      GW=(PSI-PSI2(I))/(PSI1(I)-PSI2(I))
      IF (GW.GT.1.0) GW=1.0
      IF (GW.LT.0.1) GW=0.1
      IF (GD.GT.1.0) GD=1.0
      IF (GD.LT.0.1) GD=0.1
C --  Stomatal resistance for water vapor
      RST=1.0/(GSPAR*GT*GD*GW)

      END IF
C
c   Decide if dew or rain occurs.
C
      IF (FCLD.LT.0.25) THEN
        Coedew=0.3
      ELSE  IF (FCLD.GE.0.25.AND.FCLD.LT.0.75) THEN
         Coedew=0.2
      ELSE
        Coedew=0.1
      END IF
         DQ=0.622/1000. * ES(TS)*(1.- RH)*1000.    ! unit g/kg
         DQ=amax1(0.0001,DQ)
         USMIN=1.5/DQ*Coedew

      IF (TS.GT.273.15 .AND. PREC.GT.0.20) then
        is_rain = .true.
        is_dew = .false.
      ELSE IF (TS.GT.273.15 .AND. USTAR(I).LT.USMIN)THEN
        is_dew = .true.
        is_rain = .false.
      ELSE
        is_rain = .false.
        is_dew = .false.
      END IF
C
C   Decide fraction of stomatal blocking due to wet conditions
C
      Wst=0.
      if ((is_dew.or.is_rain).and.SRAD.GT.200.) then
      Wst=(SRAD-200.)/800.
      Wst=amin1(Wst, 0.5)
```

```
      end if
C
C -- In-canopy aerodynamic resistance
C
        Rac = Rac1(I)+(LAI_F(I)-LAI(I,14))/(LAI(I,15)-LAI(I,14)+1.E-10)
     &             *(Rac2(I)-Rac1(I))
        Rac = Rac*LAI_F(I)**0.25/USTAR(I)/USTAR(I)
C
C -- Ground resistance for O3
C
      IF (I.GE.4.AND.TS.LT.272.15) THEN
        RgO_F = amin1( RgO(I)*2., RgO(I) * exp(0.2*(272.15-TS)))
      ELSE
        RgO_F = RgO(I)
      END IF
C
C -- Ground resistance for SO2
C
      IF (I.EQ.2) THEN
        RgS_F = AMIN1(RgS(I)*(275.15-TS), 500.)
        RgS_F = AMAX1(RgS(I), 100.)
      ELSE IF (I.GE.4.AND.is_rain) THEN
        RgS_F = 50.
      ELSE IF (I.GE.4.AND.is_dew) THEN
        RgS_F = 100.
      ELSE IF (I.GE.4.AND.TS.LT.272.15) THEN
        RgS_F = amin1( RgS(I)*2., RgS(I) * exp(0.2*(272.15-TS)))
      ELSE
        RgS_F =  RgS(I)
      END IF
C
C -- Cuticle resistance for O3 AND SO2
C
      IF (RcutdO(I).LE.-1) THEN
        RcutO_F = 1.E25
        RcutS_F = 1.E25
      ELSE IF (is_rain) THEN
        RcutO_F = RcutwO(I)/LAI_F(I)**0.5/USTAR(I)
        RcutS_F = 50./LAI_F(I)**0.5/USTAR(I)
        RcutS_F = MAX (RcutS_F, 20.)
      ELSE IF (is_dew) THEN
        RcutO_F = RcutwO(I)/LAI_F(I)**0.5/USTAR(I)
        RcutS_F = 100./LAI_F(I)**0.5/USTAR(I)
        RcutS_F = MAX (RcutS_F, 20.)
      ELSE IF (TS.LT.272.15) THEN
        RcutO_F = RcutdO(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I)
        RcutS_F = RcutdS(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I)
        RcutO_F = amin1( RcutO_F*2., RcutO_F * exp(0.2*(272.15-TS)))
        RcutS_F = amin1( RcutS_F*2., RcutS_F * exp(0.2*(272.15-TS)))
        RcutO_F = MAX (RcutO_F,100.)
        RcutS_F = MAX (RcutS_F,100.)
      ELSE
        RcutO_F = RcutdO(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I)
        RcutS_F = RcutdS(I)/exp(3.*RH)/LAI_F(I)**0.25/USTAR(I)
        RcutO_F = MAX (RcutO_F,100.)
        RcutS_F = MAX (RcutS_F,100.)
      END IF
```

```
C
C If snow occurs, Rg and Rcut are adjusted by snow cover fraction
C
        fsnow= sd/sdmax(i)
        fsnow= amin1(1.0, fsnow)    !snow cover fraction for leaves
      If (fsnow.GT.0.0001.and.I.GE.4) THEN
        RsnowS= AMIN1(70.*(275.15-TS), 500.)
        RsnowS= AMAX1(RSnowS, 100.)
        RcutS_F=1.0/((1.-fsnow)/RcutS_F+fsnow/RsnowS)
        RcutO_F=1.0/((1.-fsnow)/RcutO_F+fsnow/2000.)
        fsnow= amin1(1.0, fsnow*2.)    !snow cover fraction for ground
        RgS_F=1.0/((1.-fsnow)/RgS_F+fsnow/RsnowS)
        RgO_F=1.0/((1.-fsnow)/RgO_F+fsnow/2000.)
      END IF

C
C Loop 100 for gas species
C
      DO 100 J=1,NG
C
C -- Calculate diffusivity for each gas species
C
        dgas=0.369*MW(J)+6.29
        DI=0.001*TS**1.75*SQRT((29.+MW(J))/MW(J)/29.)
        DI=DI/1.0/(dair**0.3333+dgas**0.3333)**2
        VI=145.8*1.E-4*(TS*0.5+T2*0.5)**1.5/
     &                  (TS*0.5+T2*0.5+110.4)
        VI=VI/ROAROW
C
C -- Calculate quasi-laminar resistance
C
        Rb  =5./USTAR(I)*(VI/DI)**.666667
C
C -- Calculate stomatal resistance for each species from the ratio of
C       diffusity of water vapor to the gas species
C
        DVh2o=0.001*TS**1.75*SQRT((29.+18.)/29./18.)
        DVh2o=DVh2o/(dair**0.3333+dh2o**0.3333)**2
        RS=RST*DVh2o/DI+RM(J)
C
C -- Scale cuticle and ground resistances for each species
C
        Rcut = 1./(ALPHA(J)/RcutS_F+BETA(J)/RcutO_F)
        Rg = 1./(ALPHA(J)/RgS_F+BETA(J)/RgO_F)
C
C -- Calculate total surface resistance
C
        Rc = (1.-Wst)/Rs+1./(Rac+Rg)+1./Rcut
        Rc=amax1(10.0,1./Rc)         !Set minimum surface resistance as 10
s/m
C
C -- Deposition velocity
C
        VDG(I,J) = 1./(RA(I)+RB+RC)

100   CONTINUE   ! end of gaseous species
```

```
200   CONTINUE   ! end of LUC

      RETURN
      END
```